

## GiftSet Instanzen

```
module GiftSet where

import Lib (CanBeEmpty(..), FromList(..), ToList(..))

data GiftSet a = GiftSet a | EmptyGiftSet
  deriving (Eq)

-- BoxIt und UnboxIt Funktionen
boxIt :: a -> GiftSet a
boxIt = GiftSet

unboxIt :: GiftSet a -> (a, GiftSet a)
unboxIt EmptyGiftSet = error "cannot unbox an empty gift set"
unboxIt (GiftSet x) = (x, EmptyGiftSet)

-- Show-Instanz für GiftSet
instance Show a => Show (GiftSet a) where
  show EmptyGiftSet = "*** _____ ***"
  show (GiftSet x) = "***" ++ show x ++ "***"

-- Functor-Instanz für GiftSet (abgeleitet)
instance Functor GiftSet where
  fmap _ EmptyGiftSet = EmptyGiftSet
  fmap f (GiftSet x) = GiftSet (f x)

-- Applicative-Instanz für GiftSet
-- {-# MINIMAL pure, ((<*>) | liftA2) #-}
instance Applicative GiftSet where
  pure = GiftSet
  EmptyGiftSet <*> _ = EmptyGiftSet
  _ <*> EmptyGiftSet = EmptyGiftSet
  (GiftSet f) <*> (GiftSet x) = GiftSet (f x)

-- Monad-Instanz für GiftSet
-- {-# MINIMAL (>>=) #-}
instance Monad GiftSet where
  return = pure
  EmptyGiftSet >>= _ = EmptyGiftSet
  (GiftSet x) >>= f = f x
```

```

-- Foldable-Instanz für GiftSet
-- {-# MINIMAL foldMap | foldr #-}
instance Foldable GiftSet where
    foldr _ z EmptyGiftSet = z
    foldr f z (GiftSet x) = f x z

    foldl _ z EmptyGiftSet = z
    foldl f z (GiftSet x) = f z x

    foldMap _ EmptyGiftSet = mempty
    foldMap f (GiftSet x) = f x

-- Semigroup-Instanz für GiftSet
-- {-# MINIMAL (<>) #-}
instance Semigroup a => Semigroup (GiftSet a) where
    EmptyGiftSet <> r = r
    l <> EmptyGiftSet = l
    (GiftSet x) <> (GiftSet y) = GiftSet (x <> y)

-- Monoid-Instanz für GiftSet
-- {-# MINIMAL mempty #-}
instance Semigroup a => Monoid (GiftSet a) where
    mempty = EmptyGiftSet
    mappend = (<>)

-- ToList-Instanz für GiftSet
-- {-# MINIMAL toList #-}
instance ToList GiftSet where
    toList EmptyGiftSet = []
    toList (GiftSet x) = [x]

-- FromList-Instanz für GiftSet
-- {-# MINIMAL fromList #-}
instance FromList GiftSet where
    fromList [] = EmptyGiftSet
    fromList (x:_) = GiftSet x

```